

Estruturas de Controle

Seleção ou Desvio Condicional

a) Seleção Simples:

Permite a escolha de um grupo de ações (bloco) a ser executado quando determinadas condições, representadas por expressões lógicas ou relacionais são ou não satisfeitas.

Sintaxe usada:

```
se (condição) entao
    comandos
fimse
```

b) Seleção Composta:

Acontece quando em uma dada situação existem duas alternativas que dependem de uma mesma condição. Se a condição for verdadeira, um caminho será seguido, se for falsa, outro caminho será tomado.

Sintaxe usada:

```
se (condição) entao
    comandos
senao
    comandos
fimse
```

c) Estrutura de Decisão Aninhada ou Encadeada:

Existem casos em que é necessário estabelecer verificação de condições sucessivas, em que uma determinada ação poderá ser executada se um conjunto anterior de instruções ou condições for satisfeito. Sendo a ação executada, ela poderá ainda estabelecer novas condições. Isso significa utilizar uma condição dentro de outra condição. Esse tipo de estrutura poderá possuir diversos níveis de condição, sendo chamadas de aninhadas ou encadeamentos.

Sintaxe usada:

```

se <condição> entao
    comando1
senao
    se <condição2> entao
        comando2
    senao
        se <condição3> entao
            comando 3
        senao
            comando 4
        fimse
    fimse
fimse

```

O uso de estruturas de seleção encadeada envolve normalmente problemas com diversas condições. Veja a necessidade de seu uso a partir do seguinte problema: Escreva um algoritmo para imprimir o conceito final de um aluno a partir de uma média. A tabela de conversão a ser utilizada é a seguinte:

Conceito “A” se media ≥ 9	Conceito “C” se media ≥ 6
Conceito “B” se media ≥ 7	Conceito “P” se media < 6

Aplicando a técnica de interpretação de enunciado:

SAÍDA:

Quais os resultados que o algoritmo deve fornecer?

Saída: Conceito final

Próximo passo **ENTRADA:**

Quais os insumos necessários para se obter a saída?

Precisa conhecer a média do aluno. Dá para supor que a media será fornecida automaticamente pelo programa, mas vamos fazer que seja fornecida pelo usuário diretamente.

ENTRADA de DADOS:

media

Passos para obter os insumos necessários para saída de dados:

Conceito “A” se $media \geq 9$;

Conceito “B” se $media \geq 7$;

Conceito “C” se $media \geq 6$;

Conceito “P” se $media < 6$;

Estrutura de seleção:

```
se (media  $\geq$  9) entao
    escreva(“Conceito A”)
fimse
```

```
se (media  $\geq$  7) entao
    escreva(“Conceito B”)
fimse
```

```
se (media  $\geq$  6) entao
    escreva(“Conceito C”)
fimse
```

```
se (media  $<$  6) entao
    escreva(“Conceito P”)
fimse
```

Simulando a execução do programa teremos o seguinte:

Declaração de Variáveis

media: real

Instruções de Entrada de dados:

escreva("Digite a media: ")

leia(media)

Na sequencia temos as estrutura de seleção com os testes de condição.

se (media \geq 9) entao

 escreva("Conceito A")

fimse

se (media \geq 7) entao

 escreva("Conceito B")

fimse

se (media \geq 6) entao

 escreva("Conceito C")

fimse

se (media $<$ 6) entao

 escreva("Conceito P")

fimse

Simulando a execução do programa quais conceitos obteriam como saída, se o usuário digitar valor 9.5 para media?

A resposta teria que ser Conceito A, no entanto a solução imprime três mensagens:

Conceito A

Conceito B resultados de saída.
Conceito C

Assim executa os três primeiros blocos de instruções devidas serem verdadeiro. Isso ocorre devido ao uso de quatro estruturas de seleção simples ao invés de seleção composta com senão. Outra possibilidade seria fazer uso de seleção composta com operadores lógicos verificando os intervalos. Por exemplo:

```
se (media >= 9) entao
    escreva("Conceito A")
fimse
```

```
se ((media >= 7) e (media < 9)) entao
    escreva("Conceito B")
fimse
```

```
se ((media >= 6) e (media < 7)) entao
    escreva("Conceito C")
fimse
```

```
se (media < 6) entao
    escreva("Conceito P")
fimse
```

Com essa estrutura somente o conceito "A" será exibido o qual está correto. No entanto essa estrutura ainda não é conveniente. Ela é eficaz mas ainda não é ideal. Nesse caso como temos vários SEs, precisamos da estrutura de seleção aninhada ou encadeada.

ESTRUTURAS DE SELEÇÃO ENCADEADAS são sequências de estruturas SE...ENTÃO...SENÃO... usados para testes de múltiplos casos encadeados.

```
se (media >= 9) então
    escreva("Conceito A")
senão
    se (media >= 7) então
        escreva("Conceito B")
    senão
        se (media >= 6) então
            escreva("Conceito C")
        senão
            escreva("Conceito P")
        fimse
    fimse
fimse
```

Exercícios sobre estrutura de decisão aninhada ou encadeada:


1) Elaborar um algoritmo que verifique qual dos dois valores digitado pelo usuário é maior, ou se os mesmos são iguais. Observem o exercício resolvido na figura 1

```

1 Algoritmo "Exercício1"
2
3 Var
4   n1, n2: inteiro
5 Inicio
6
7 escreva("Informe um valor: ")
8 leia(n1)
9 escreva("Informe o segundo valor: ")
10 leia(n2)
11
12 se (n1>n2) entao
13     escreva(n1," é maior que ",n2)
14 senao
15     se (n2>n1) entao
16         escreva(n2," é maior que ",n1)
17     senao
18         escreva("Os valores são iguais ")
19     fimse
20 fimse
21
22 Fimalgoritmo

```

Figura 1 - Exercício 1 em português estruturado



```

C:\ Console simulando o modo texto do MS-DOS

Informe um valor: 5
Informe o segundo valor: 7
 7 é maior que 5
>>> Fim da execução do programa !

```

Figura 2 - Exercício 1, console, entrada e saída de dados

Após a execução (F9) do algoritmo no VISUALG, conforme ilustrado na figura 1, com a entrada dos dois valores nas variáveis n1, n2, faz uma verificação na primeira condição (n1>n2), que se lê, o valor de n1 é maior que o valor da variável n2. Na figura 2, observem que o valor que a variável n1 recebe é o valor 5 e a variável n2 valor 7. Na execução do algoritmo, a leitura das linhas de códigos se dá de cima para baixo, dá esquerda para direita, linha após linha, de forma sequencial. E se tratando de estrutura de seleção, a partir disso, faz com que execute ou não um

determinado bloco de código. Na linha 12 da figura 1, a primeira condição resulta em falso, por que o valor 5 não é maior que o valor 7. Isso faz que não entre no primeiro "se" e pule para a próxima condição ($n2 > n1$) na linha 15 da figura 1, retornando verdadeiro, por que 7 é maior que 5 e entra nesse bloco e executa o comando `escreva(n2, " é maior que ", n1)`. Após executar o comando a leitura do código pula para o final do fimse.

2) Elaborar um algoritmo que lê um número e identifica se ele é positivo, negativo ou nulo.

Obs.: A seguir, estrutura de decisão utilizando operadores lógicos:

3) Elaborar um algoritmo que lê três números e escrever o maior. Observem o exercício resolvido na figura 3.

```
1 Algoritmo "exercício3"
2
3 Var
4
5 n1, n2, n3, acumulador: inteiro
6
7 Inicio
8     escreva("Informe 3 números inteiros: ")
9     leia(n1,n2,n3)
10    se ((n1>=n2) e (n1>=n3)) entao
11        acumulador := n1
12    senao
13        se ((n2>=n1) e (n2>=n3)) entao
14            acumulador := n2
15        senao
16            acumulador := n3
17
18        fimse
19    fimse
20    escreva("O maior número é ",acumulador)
21 Fimalgoritmo
```

Figura 3 – Exercício 3 em português estruturado

C:\ Console simulando o modo texto do MS-DOS

```
Informe 3 números inteiros: 5
9
2
O maior número é 9
>>> Fim da execução do programa !
```

Figura 4 - Exercício 3, console, entrada e saída de dados

Conforme figura 3, na linha 10 onde tem a primeira verificação referente aos valores das variáveis (($n1 \geq n2$) e ($n1 \geq n3$)), usando como referencia para essa explicação a figura 4, observem que a variável **n1** recebe o valor 5, **n2** o valor 9 e **n3** o valor 2, vai retornar o valor falso, ou seja, não entra nesse bloco. Por que retorna falso? Como as expressões estão ligadas pelo operador lógico " e " e substituindo na representação (($n1 \geq n2$) e ($n1 \geq n3$)) pelos valores de cada variável pelo respectivo algarismo que receberam na entrada dos dados:

(($5 \geq 9$) e ($5 \geq 2$))

falso verdadeiro

falso e verdadeiro = falso

Quando se usa operador lógico E, essa conjunção só vai ser verdadeira se todas as expressões forem verdadeiras. Basta uma ser falsa para retornar falso. O operador lógico E, é somatório, todas as expressões matemáticas tem que serem verdadeiras para o resultado final ser verdadeiro.

Por exemplo, se trocarmos a primeira condição ($5 \geq 9$) pela proposição A e a segundo ($5 \geq 2$) pela proposição B.

Vou usar essas proposições (A e B) em uma tabela-verdade para representar graficamente a saída de dados, ou resultado:

Como substitui as 2 expressões pelas proposições A e B, para fazer a tabela-verdade, vai ser $2^2 = 2 \times 2 = 4$, quatro linha na tabela. O que isso significa? Significa base 2 os (0 e 1) em base binária, ou V e F, são os valores que trabalhamos, não existe um terceiro valor.

Ou representamos por 0 e 1 ou podemos representar esses valores com V ou F (Verdadeiro ou Falso).

E o valor no expoente 2? Representa o número de proposições que envolvem na condição.

A	B	A e B
V	V	V
V	F	F
F	V	F
F	F	F

Observem que as duas condições tem que ser verdadeiras para termos um resultado verdadeiro. Conforme a primeira condição analisada do exercício 3, $((5 \geq 9) \text{ e } (5 \geq 2))$ que resulta em falso, podemos observar na tabela-verdade acima que está representado, na linha 3 em laranja. Essas quatro linhas da tabela-verdade representam os possíveis resultados a partir dos valores de entrada das variáveis $(n1, n2)$.

Se trocarmos o operador lógico E por OU como fica com base no exemplo acima?
 $((5 \geq 9) \text{ ou } (5 \geq 2))$

falso verdadeiro = verdadeiro

A	B	A ou B
V	V	V
V	F	V
F	V	V
F	F	F

Observem que agora basta uma das proposições ser verdadeira para o resultado ser verdadeiro. Operador OU, o próprio nome sugere, ou um ou outro já basta para resultar em verdadeiro.

ATENÇÃO: Aconselho ler na apostila algoritmo.pdf, ou fazerem uma pesquisa na internet sobre tabela-verdade para um melhor entendimento de condições envolvendo operadores lógicos.

4) Construir um algoritmo que leia duas notas e calcule a média final. Após o cálculo, escreva o resultado final de acordo com os dados:

condição	resultado
$\text{media} \geq 7$	aprovado
$\text{media} \geq 5 \text{ e } \text{media} < 7$	exame
$\text{media} < 5$	reprovado

5) Algoritmo que lê três notas de um aluno e imprimir a media com os dados de acordo com os dados abaixo:

Condição	resultado
$\text{media} \geq 8$	Ótimo
$\text{media} \geq 7 \text{ e } \text{media} < 8$	Bom
$\text{media} \geq 5 \text{ e } \text{media} < 7$	Regular
$\text{media} < 5$	Péssimo

6) Elaborar um algoritmo para uma calculadora básica. Ler dois valores e perguntar qual das operações básicas matemática irá ser realizada.

7) Algoritmo que verifique se um número digitado pelo usuário está na faixa dos 20 e 90.